Computação Algébrica no Cálculo das Variações: Determinação de Simetrias e Leis de Conservação*

Paulo D. F. Gouveia
pgouveia@ipb.pt

Delfim F. M. Torres
delfim@mat.ua.pt

Control Theory Group (cotg)
Centre for Research in Optimization and Control (CEOC)
University of Aveiro, Department of Mathematics
3810-193 Aveiro, Portugal

Resumo. Os problemas de optimização dinâmica (em espaços de funções) tratados pelo cálculo das variações, são normalmente resolvidos por recurso às condições necessárias de Euler-Lagrange, que são equações diferenciais de segunda ordem (ou de ordem superior, quando os problemas variacionais envolvem derivadas de ordem superior a um). Estas equações são, em geral, não lineares e de difícil resolução. Uma forma de as simplificar consiste em obter leis de conservação: primeiros integrais das equações diferenciais de Euler-Lagrange. Os primeiros integrais permitem baixar a ordem das equações e, em casos extremos, com um número suficientemente grande de primeiros integrais independentes, resolver o problema por completo. Se em áreas como a Física e a Economia a questão da existência de leis de conservação é resolvida de forma bastante natural, a própria aplicação sugerindo as leis de conservação (e.g. conservação de energia, conservação da quantidade de movimento, conservação do rendimento, etc.), de um ponto de vista estritamente matemático, dado um problema do cálculo de variações, o processo de obtenção das leis de conservação ou, até mesmo, a demonstração de que elas existem (ou não), deixa de ser uma questão óbvia. Neste trabalho mostramos como um sistema de computação algébrica como o Maple pode ser muito útil na abordagem a estas questões. Apresentamos um conjunto de facilidades computacionais simbólicas que permitem, de uma forma sistemática e automática, identificar as leis de conservação de uma dada funcional integral do cálculo das variações. O algoritmo usado tem por base o célebre teorema de Emmy Noether, que associa a existência de leis de conservação às propriedades de invariância do problema (à existência de simetrias variacionais). Vários exemplos ilustrativos são apresentados, mostrando a utilidade das ferramentas desenvolvidas.

Abstract. The dynamic optimization problems treated by the calculus of variations are usually solved with the help of the 2nd order Euler-Lagrange differential

*Partially presented at XXVII CNMAC (Congresso Nacional de Matemática Aplicada e Computacional), FAMAT/PUCRS, Porto Alegre, RS, Brasil, 13-16 September 2004. Supported by the program PRODEP III/5.3/2003 and the R&D unit CEOC.

equations. These equations are, generally speaking, nonlinear, and very hard to solve. One way to address the problem is to obtain conservation laws of lower order than those of the corresponding Euler-Lagrange equations. While in Physics and Economics the question of existence of conservation laws is treated in a rather natural way, because the application itself suggest the conservation laws (e.g., conservation of energy, income/health law), from a strictly mathematical point of view, given a problem of the calculus of variations, it is not obvious how one might derive a conservation law or, for that matter, if it even has a conservation law. The question we address is thus to develop computational facilities, based on a systematic method, which permits to identify functionals that have conservation laws. The central result we use is the celebrated Noether's theorem. This theorem links conservation laws with the invariance properties of the problem (with symmetries), and provides an algorithm for finding conservation laws. Thus the problem is reduced to the one of finding the variational symmetries. We show how a Computer Algebra System can help to find the symmetries and the conservation laws in the calculus of variations. Several illustrative examples are given.

Mathematics Subject Classification 2000: 49-04, 49K05, 49S05.

1. Introdução

O cálculo das variações é uma área clássica da matemática, com mais de três séculos de idade, extremamente activa no século XXI, e com inúmeras aplicações práticas na mecânica, economia, ciências dos materiais, ciências do espaço e engenharia. O cálculo das variações está na origem de muitas áreas mais recentes, como sejam a análise funcional e o controlo óptimo [4, 7, 8].

A computação algébrica, também chamada de computação científica ou simbólica, permite trabalhar com expressões matemáticas de maneira simbólica, não numérica, e é uma área de investigação moderna, que surgiu na segunda metade do século XX. Se é verdade que os actuais sistemas de computação algébrica são extremamente poderosos, também não é menos verdade que muito existe por fazer no sentido de se colocarem os computadores a realizar tarefas matemáticas verdadeiramente interessantes.

O uso da teoria do cálculo das variações, na resolução de problemas concretos, requer cálculos não numéricos: no cálculo das variações a presença de fenómenos como o de Lavrentiev [3], faz com que as soluções algébricas exactas sejam muito mais convenientes do que as soluções numéricas. Tendo em conta que a resolução dos problemas do cálculo das variações passa quase obrigatoriamente pela resolução das equações diferenciais de Euler-Lagrange, equações estas, em geral, de difícil resolução, neste trabalho propomos um conjunto de procedimentos computacionais algébricos que permitem automatizar o processo de obtenção de leis de conservação (primeiros integrais das equações de Euler-Lagrange). Como é bem conhecido da teoria das equações diferenciais, estas leis de conservação são de extrema utilidade, permitindo baixar a ordem das equações [12].

O presente trabalho encontra-se organizado em sete secções. Em §2. fazemos uma breve apresentação do cálculo das variações, da condição necessária de Euler-

Lagrange (definida em Maple na §6.) e definimos *lei de conservação*. Em §3. definimos *simetria* (*invariância* de um problema do cálculo das variações) e apresentamos um método construtivo para as obter (implementado em Maple na §6., por intermédio do procedimento *Simetria*). Em §4. formulamos o célebre princípio de Noether, que nos dá uma fórmula explícita para as leis de conservação do cálculo das variações, em função dos geradores que definem a simetria. O teorema de Noether é definido em Maple na §6. por intermédio do procedimento *Noether*. Vários exemplos, mostrando o modo de uso e a utilidade das funções Maple definidas em §6., são apresentados em §5.. Por fim, apresentamos em §7. algumas conclusões e perspectivas de trabalho futuro.

2. Cálculo das Variações

Neste trabalho consideramos problemas do cálculo das variações de ordem superior: minimizar uma funcional integral

$$J[\mathbf{x}(\cdot)] = \int_a^b L(t, \mathbf{x}(t), \dot{\mathbf{x}}(t), \ldots, \mathbf{x}^{(m)}(t))\, \mathrm{d}t\,, \tag{2.1}$$

onde o Lagrangeano L é uma função real que assumimos ser continuamente diferenciável em $[a,b] \times \mathbb{R}^{n\times(m+1)}$; $t \in \mathbb{R}$, é a variável independente; $\mathbf{x}(t) = [x_1(t)\ x_2(t) \cdots x_n(t)]^{\mathrm{T}} \in \mathbb{R}^n$, as variáveis dependentes; $\mathbf{x}^{(i)}(t) = [\frac{\mathrm{d}^i x_1(t)}{\mathrm{d}t^i}\ \frac{\mathrm{d}^i x_2(t)}{\mathrm{d}t^i} \cdots \frac{\mathrm{d}^i x_n(t)}{\mathrm{d}t^i}]^{\mathrm{T}} \in \mathbb{R}^n$, com $i = 1, \ldots, m$, as derivadas de ordem i das variáveis dependentes em ordem a t; e $\dot{\mathbf{x}}(t) \equiv \mathbf{x}^{(1)}(t)$. Assim, o Lagrangeano considerado, depende explicitamente de uma variável independente, de n variáveis dependentes e das suas m primeiras derivadas. Para $m = 1$ obtemos o problema fundamental do cálculo das variações.

Na minimização da funcional (2.1) é usual recorrer-se ao sistema de equações de Euler-Lagrange

$$\frac{\partial L}{\partial \mathbf{x}} + \sum_{i=1}^{m} (-1)^i \frac{\mathrm{d}^i}{\mathrm{d}t^i}\left(\frac{\partial L}{\partial \mathbf{x}^{(i)}}\right) = \mathbf{0}, \tag{2.2}$$

onde $\frac{\partial L}{\partial \mathbf{x}^{(i)}} = [\frac{\partial L}{\partial x_1^{(i)}}\ \frac{\partial L}{\partial x_2^{(i)}}\ \cdots\ \frac{\partial L}{\partial x_n^{(i)}}]$ e L, e suas derivadas, são avaliadas ao longo de $(t, \mathbf{x}(t), \dot{\mathbf{x}}(t), \ldots, \mathbf{x}^{(m)}(t))$.

Definição 2.1. *Às soluções da equação de Euler-Lagrange (2.2) chamamos* extremais.

Observação 2.1. *Para o problema fundamental do cálculo das variações, i.e., quando a funcional não depende de derivadas de* $\mathbf{x}(t)$ *de maior ordem do que a primeira* ($m = 1$)*, o sistema de equações de Euler-Lagrange (2.2) reduz-se a*

$$\frac{\partial L}{\partial \mathbf{x}} - \frac{\mathrm{d}}{\mathrm{d}t}\left(\frac{\partial L}{\partial \dot{\mathbf{x}}}\right) = \mathbf{0}\,.$$

Observação 2.2. *Se a funcional não envolver mais do que uma variável dependente* ($n = 1$)*, todas as grandezas presentes em (2.1) e (2.2) são escalares.*

Em §6. definimos o procedimento *EulerLagrange* que tem por entrada o Lagrangeano e como saída o sistema de equações de Euler-Lagrange correspondente, que resulta da aplicação da equação (2.2).

Definição 2.2. *Uma função $t \to \phi(t, \mathbf{x}(t), \dot{\mathbf{x}}(t), \ldots, \mathbf{x}^{(k)}(t))$, $k < 2m$, que se mantenha constante ao longo de todas as extremais do problema (2.1), é chamada de* primeiro integral *(de ordem k) da equação de Euler-Lagrange (2.2). À equação*

$$\phi(t, \mathbf{x}(t), \dot{\mathbf{x}}(t), \ldots, \mathbf{x}^{(k)}(t)) = const$$

chamamos lei de conservação *(de ordem k).*

As leis de conservação são muito úteis, pois permitem reduzir a ordem das equações diferenciais de Euler-Lagrange. Com um número suficientemente elevado de primeiros integrais independentes, é mesmo possível determinar explicitamente as extremais. Consideremos, a título de exemplo, o seguinte problema ($n = m = 1$):

$$J[x(\cdot)] = \int_a^b \left(\dot{x}^2(t) - x^2(t)\right) \mathrm{d}t \longrightarrow \min .$$

Neste caso o Lagrangeano é dado por $L(x, \dot{x}) = -x^2 + \dot{x}^2$ e a equação de Euler-Lagrange (2.2) reduz-se a $\ddot{x}(t) + x(t) = 0$. Se multiplicarmos esta equação diferencial por $\cos(t)$ obtemos $\frac{d}{dt}(\dot{x}\cos(t) + x\sin(t)) = 0$; enquanto que se a multiplicarmos por $-\sin(t)$ obtemos $\frac{d}{dt}(-\dot{x}\sin(t) + x\cos(t)) = 0$. Temos então duas leis de conservação:

$$\begin{cases} \dot{x}\cos(t) + x\sin(t) & = c_1 , \\ -\dot{x}\sin(t) + x\cos(t) & = c_2 . \end{cases} \tag{2.3}$$

Resulta de imediato das duas leis de conservação (2.3) que as extremais têm a forma $x(t) = c_1 \sin(t) + c_2 \cos(t)$. Claro que este exemplo é trivial: a equação diferencial de Euler-Lagrange pode facilmente ser resolvida sem recurso a leis de conservação, uma vez que ela é linear. Usando o Maple e o nosso procedimento *EulerLagrange* faríamos:

```
> L := v^2 - x^2:
> dsolve(EulerLagrange(L,t,x,v));
```

$$\{x(t) = \_C1 \sin(t) + \_C2 \cos(t)\}$$

No caso não linear o comando Maple *dsolve* nem sempre é capaz de obter soluções explícitas (e.g. para a equação não linear de Emden-Fowler considerada no Exemplo 5.4; ou para o problema de Kepler tratado no Exemplo 5.2). Nessas situações as leis de conservação são muito úteis. Coloca-se então a seguinte questão: *Dada uma funcional integral do tipo (2.1), como obter leis de conservação?* Esta questão foi resolvida por Emmy Noether em 1918 [11]: se a funcional for invariante sob determinado tipo de transformações (transformações de invariância ou simetrias), então existem fórmulas explícitas para as leis de conservação. O Teorema de Noether encontra muitas aplicações em campos concretos da Engenharia, como a Sismologia e a Metalurgia [6]. A dificuldade na sua aplicação reside na obtenção das simetrias (na obtenção das transformações de invariância). Neste trabalho automatizamos, por recurso ao sistema de computação algébrica Maple [1], a determinação de simetrias e a correspondente aplicação do Teorema de Noether.

3. Simetrias Variacionais

Para o estudo das propriedades de invariância das funcionais do cálculo das variações considera-se uma família uni-paramétrica de transformações $\mathbf{h}^s(t, \mathbf{x})$ que forma um grupo local de Lie [10, 2]. A família uni-paramétrica $\mathbf{h}^s(t, \mathbf{x})$ representa um conjunto de $n+1$ transformações de $[a, b] \times \mathbb{R}^n$ em $\mathbb{R} \times \mathbb{R}^n$

$$t^s = h_t^s(t, \mathbf{x}), \quad x_i^s = h_{x_i}^s(t, \mathbf{x}), \text{ com } i = 1, \ldots, n, \tag{3.1}$$

a que correspondem $n+1$ geradores infinitesimais representados por

$$T(t, \mathbf{x}) = \left.\frac{\partial}{\partial s} h_t^s(t, \mathbf{x})\right|_{s=0}, \quad X_i(t, \mathbf{x}) = \left.\frac{\partial}{\partial s} h_{x_i}^s(t, \mathbf{x})\right|_{s=0}, \text{ com } i = 1, \ldots, n. \tag{3.2}$$

Definição 3.1. *A funcional (2.1) diz-se invariante no intervalo $[a, b]$ sob as transformações uni-paramétricas (3.1) se, para todo o s suficientemente pequeno,*

$$\int_\alpha^\beta L(t, \mathbf{x}(t), \dot{\mathbf{x}}(t), \ldots, \mathbf{x}^{(m)}(t))\, \mathrm{d}t = \int_{\alpha^s}^{\beta^s} L(t^s, \mathbf{x}^s(t^s), \dot{\mathbf{x}}^s(t^s), \ldots, \mathbf{x}^{s(m)}(t^s))\, \mathrm{d}t^s,$$

em qualquer subintervalo $[\alpha, \beta] \subseteq [a, b]$; com $\alpha^s = h_t^s(\alpha, \mathbf{x}(\alpha))$ e $\beta^s = h_t^s(\beta, \mathbf{x}(\beta))$.

Observação 3.1. *Nas condições da definição 3.1 as transformações uni-paramétricas (3.1) constituem uma* simetria variacional *da funcional (2.1).*

O teorema que se segue estabelece uma condição necessária e suficiente de invariância, de extrema importância para os objectivos a que nos propomos.

Teorema 3.1 ([15]). *A funcional (2.1) é invariante sob as transformações uni-paramétricas (3.1), com geradores infinitesimais T e $\mathbf{X}$ (3.2), se, e apenas se,*

$$\frac{\partial L}{\partial t} T + \sum_{i=0}^{m} \frac{\partial L}{\partial \mathbf{x}^{(i)}} \cdot \mathbf{p}^i + L \frac{\mathrm{d}T}{\mathrm{d}t} = 0, \tag{3.3}$$

onde

$$\mathbf{p}^0 = \mathbf{X}, \quad \mathbf{p}^{i+1} = \frac{\mathrm{d}\mathbf{p}^i}{\mathrm{d}t} - \mathbf{x}^{(i+1)} \frac{\mathrm{d}T}{\mathrm{d}t}, \ i = 0, \ldots, m-1\,. \tag{3.4}$$

Em (3.3) e (3.4) assumimos que T e $\mathbf{X} = [X_1 \ X_2 \cdots X_n]^T$ são avaliadas em função de $(t, \mathbf{x})$ e $\mathbf{p}^i$ em função de $(t, \mathbf{x}, \dot{\mathbf{x}}, \ldots, \mathbf{x}^{(i)})$, com $i = 0, 1, \ldots, m$.

Corolário 3.1.1. *Quando o Lagrangeano L não depende de derivadas de $\mathbf{x}(t)$ de ordem superior à primeira ($m = 1$), a equação (3.3) toma a forma*

$$\frac{\partial L}{\partial t}T + \frac{\partial L}{\partial \mathbf{x}} \cdot \mathbf{X} + \frac{\partial L}{\partial \dot{\mathbf{x}}} \cdot \left(\frac{\mathrm{d}\mathbf{X}}{\mathrm{d}t} - \dot{\mathbf{x}}\frac{\mathrm{d}T}{\mathrm{d}t}\right) + L\frac{\mathrm{d}T}{\mathrm{d}t} = 0,$$

com

$$\frac{dT}{dt} = \frac{\partial T}{\partial t} + \frac{\partial T}{\partial \mathbf{x}} \cdot \dot{\mathbf{x}}, \quad \frac{\mathrm{d}\mathbf{X}}{\mathrm{d}t} = \frac{\partial \mathbf{X}}{\partial t} + \frac{\partial \mathbf{X}}{\partial \mathbf{x}} \cdot \dot{\mathbf{x}},$$

onde

$$\frac{\partial \mathbf{X}}{\partial \mathbf{x}} = \begin{bmatrix} \frac{\partial \mathbf{X}}{\partial x_1} & \frac{\partial \mathbf{X}}{\partial x_2} & \cdots & \frac{\partial \mathbf{X}}{\partial x_n} \end{bmatrix} = \begin{bmatrix} \frac{\partial X_1}{\partial x_1} & \frac{\partial X_1}{\partial x_2} & \cdots & \frac{\partial X_1}{\partial x_n} \\ \frac{\partial X_2}{\partial x_1} & \frac{\partial X_2}{\partial x_2} & \cdots & \frac{\partial X_2}{\partial x_n} \\ \vdots & \vdots & \ddots & \vdots \\ \frac{\partial X_n}{\partial x_1} & \frac{\partial X_n}{\partial x_2} & \cdots & \frac{\partial X_n}{\partial x_n} \end{bmatrix}.$$

Observação 3.2. *Todas as derivadas totais presentes em (3.3) e (3.4) podem ser expressas por derivadas parciais, usando as igualdades $\frac{dT}{dt} = \frac{\partial T}{\partial t} + \frac{\partial T}{\partial \mathbf{x}} \cdot \dot{\mathbf{x}}$ e*

$$\frac{d\mathbf{p}^i}{dt} = \frac{\partial \mathbf{p}^i}{\partial t} + \sum_{k=0}^{i} \frac{\partial \mathbf{p}^i}{\partial \mathbf{x}^{(k)}} \cdot \mathbf{x}^{(k+1)}, \quad i = 0, \ldots, m-1,$$

onde

$$\frac{\partial \mathbf{p}^i}{\partial \mathbf{x}^{(k)}} = \begin{bmatrix} \frac{\partial \mathbf{p}^i}{\partial x_1^{(k)}} & \frac{\partial \mathbf{p}^i}{\partial x_2^{(k)}} & \cdots & \frac{\partial \mathbf{p}^i}{\partial x_n^{(k)}} \end{bmatrix} = \begin{bmatrix} \frac{\partial p_1^i}{\partial x_1^{(k)}} & \frac{\partial p_1^i}{\partial x_2^{(k)}} & \cdots & \frac{\partial p_1^i}{\partial x_n^{(k)}} \\ \frac{\partial p_2^i}{\partial x_1^{(k)}} & \frac{\partial p_2^i}{\partial x_2^{(k)}} & \cdots & \frac{\partial p_2^i}{\partial x_n^{(k)}} \\ \vdots & \vdots & \ddots & \vdots \\ \frac{\partial p_n^i}{\partial x_1^{(k)}} & \frac{\partial p_n^i}{\partial x_2^{(k)}} & \cdots & \frac{\partial p_n^i}{\partial x_n^{(k)}} \end{bmatrix}.$$

O teorema 3.1, para além de servir de teste à existência de simetrias, estabelece um algoritmo para a determinação dos correspondentes geradores infinitesimais. Como veremos, este facto é crucial: o teorema de Noether (Teorema 4.1) afirma que as leis de conservação associadas a uma dada simetria variacional apenas dependem dos geradores infinitesimais.

Dado um Lagrangeano L, determinamos os geradores infinitesimais T e $\mathbf{X}$ de uma família uni-paramétrica de transformações simétricas pelo seguinte método. A equação (3.3) é uma equação diferencial nas $n+1$ funções incógnitas T, X_1, X_2, …, e X_n, que pretendemos determinar. Porém, a equação tem de permanecer válida para todos os x_i, $i = 1, \ldots, n$. Como as funções T, X_1, X_2, …, e X_n dependem de t e x_i, $i = 1, \ldots, n$, ao substituirmos, na equação (3.3), L e todas as suas derivadas parciais pelos seus valores, obtemos um polinómio nas $n \times m$ variáveis $\dot{x}_1, \ldots, \dot{x}_n$, $x_1^{(2)}, \ldots, x_n^{(2)}, \ldots, x_1^{(m)}, \ldots, x_n^{(m)}$ e suas potências. Para que a equação seja válida para todos os valores das variáveis do polinómio, todos os seus coeficientes devem ser nulos. Notamos que os termos do polinómio poderão ser em maior número que as incógnitas do problema $(n+1)$, pelo que a condição necessária e suficiente (3.3) pode conduzir a um sistema de equações sem solução. Tal facto significa apenas que nem todas as funcionais integrais do cálculo das variações admitem simetrias variacionais (ver Exemplo 5.5). O sistema de equações a resolver, para a obtenção dos geradores, é um sistema de equações diferenciais às derivadas parciais. No entanto, ao contrário das equações diferenciais ordinárias de Euler-Lagrange, em geral não lineares e de difícil resolução, este sistema é linear em $\frac{\partial T}{\partial t}$, $\frac{\partial T}{\partial \mathbf{x}}$, $\frac{\partial \mathbf{X}}{\partial t}$ e $\frac{\partial \mathbf{X}}{\partial \mathbf{x}}$.

A resolução do sistema de equações diferencias parciais que deriva da expressão (3.3), nomeadamente quando lidamos com valores de n e m superiores à unidade, envolve um número muito elevado de cálculos, o que torna premente a necessidade de nos munirmos de ferramentas computacionais que automatizem o trabalho. Com esse fim, desenvolvemos um procedimento em `Maple`, designado *Simetria*. O respectivo código, assim como o dos restantes procedimentos por nós desenvolvidos, é apresentado na secção 6.. O procedimento *Simetria* tem por entrada a expressão que caracteriza o Lagrangeano e como saída os respectivos geradores infinitesimais. No caso do Lagrangeano não admitir qualquer simetria, obtemos geradores nulos (cf. Exemplo 5.5).

Na secção que se segue mostramos como os geradores, obtidos por intermédio do nosso procedimento *Simetria*, podem ser usados na obtenção explícita de leis de conservação.

4. Leis de Conservação

Emmy Noether foi a primeira a estabelecer a relação entre a existência de simetrias e a existência de leis de conservação. Esta ligação constitui um princípio universal, passível de ser formulado na forma de teorema nos mais diversos contextos e sob as mais variadas hipóteses [6, 9, 12, 13, 14, 15].

Teorema 4.1 (Teorema de Noether [15]). *Se a funcional (2.1) é invariante sob as transformações uni-paramétricas (3.1), com geradores infinitesimais T e $\mathbf{X}$, então*

$$\sum_{i=1}^{m} \Psi^i \cdot \mathbf{p}^{i-1} + \left(L - \sum_{i=1}^{m} \Psi^i \cdot \mathbf{x}^{(i)} \right) T = \; const, \quad t \in [a,b], \tag{4.1}$$

com

$$\begin{aligned} \Psi^m &= \frac{\partial L}{\partial \mathbf{x}^{(m)}}, \\ \Psi^{i-1} &= \frac{\partial L}{\partial \mathbf{x}^{(i-1)}} - \frac{\mathrm{d}\Psi^i}{\mathrm{d}t}, \quad i = m, m-1, \ldots, 2 \\ \frac{\mathrm{d}\Psi^i}{\mathrm{d}t} &= \frac{\partial \Psi^i}{\partial t} + \sum_{k=0}^{2m-i} \left(\mathbf{x}^{(k+1)} \right)^{\mathrm{T}} \cdot \frac{\partial \Psi^i}{\partial \mathbf{x}^{(k)}}, \end{aligned}$$

onde

$$\frac{\partial \Psi^i}{\partial \mathbf{x}^{(k)}} = \begin{bmatrix} \frac{\partial \Psi^i}{\partial x_1^{(k)}} \\ \frac{\partial \Psi^i}{\partial x_2^{(k)}} \\ \vdots \\ \frac{\partial \Psi^i}{\partial x_n^{(k)}} \end{bmatrix} = \begin{bmatrix} \frac{\partial \psi_1^i}{\partial x_1^{(k)}} & \frac{\partial \psi_2^i}{\partial x_1^{(k)}} & \cdots & \frac{\partial \psi_n^i}{\partial x_1^{(k)}} \\ \frac{\partial \psi_1^i}{\partial x_2^{(k)}} & \frac{\partial \psi_2^i}{\partial x_2^{(k)}} & \cdots & \frac{\partial \psi_n^i}{\partial x_2^{(k)}} \\ \vdots & \vdots & \ddots & \vdots \\ \frac{\partial \psi_1^i}{\partial x_n^{(k)}} & \frac{\partial \psi_2^i}{\partial x_n^{(k)}} & \cdots & \frac{\partial \psi_n^i}{\partial x_n^{(k)}} \end{bmatrix}$$

e onde se considera a grandeza Ψ^i avaliada em $(t, \mathbf{x}(t), \dot{\mathbf{x}}(t), \ldots, \mathbf{x}^{(2m-i)}(t))$, $i = 1, \ldots, m$.

Corolário 4.1.2. *Quando o Lagrangeano L não depende das derivadas de $\mathbf{x}(t)$ de maior ordem do que a primeira ($m = 1$), a equação (4.1) reduz-se a*

$$\frac{\partial L}{\partial \dot{\mathbf{x}}} \cdot \mathbf{X} + \left(L - \frac{\partial L}{\partial \dot{\mathbf{x}}} \cdot \dot{\mathbf{x}} \right) T = \text{ const.}$$

As leis de conservação que procuramos são obtidas substituindo nas equações (3.4) e (4.1) os geradores infinitesimais T e $\mathbf{X}$ encontrados pelo método descrito na secção anterior. Na secção 6. definimos o procedimento *Noether*. Este procedimento tem por entradas o Lagrangeano e os geradores infinitesimais, que são obtidos por intermédio do nosso procedimento *Simetria*, e como saída a correspondente lei de conservação (4.1). Resumindo: dado um problema do cálculo das variações (2.1), obtemos as leis de conservação, de uma forma automática, através de um processo de duas etapas: com o nosso procedimento *Simetria* obtemos todas as possíveis simetrias do problema; recorrendo depois ao nosso procedimento *Noether*, baseado no teorema 4.1, obtemos as correspondentes leis de conservação. Na secção seguinte apresentamos alguns exemplos que ilustram todo o processo.

5. Exemplos Ilustrativos

Consideramos agora várias situações concretas, mostrando a funcionalidade e a utilidade das ferramentas desenvolvidas.

Exemplo 5.1. Começamos com um exemplo muito simples em que o Lagrangeano depende apenas de uma variável dependente ($n = 1$) e não existem derivadas de ordem superior à primeira ($m = 1$): $L(t, x, \dot{x}) = t\dot{x}^2$.
Com a definição `Maple`

```
> L:= t*v^2;
```

$$L := tv^2$$

o nosso procedimento *Simetria* determina os geradores infinitesimais das simetrias do problema do cálculo das variações em consideração:

```
> Simetria(L,t,x,v);
```

$$\{T(t,x) = (2\_C1\ \ln(t) + \_C3)\,t,\ X(t,x) = \_C1\ x + \_C2\}$$

A família de geradores depende de três parâmetros que advêm das constantes de integração. Com as substituições[1]

```
> subs(_C1=1,_C2=0,_C3=0,%);
```

$$\{T(t,x) = 2\ \ln(t)\,t, X(t,x) = x\}$$

obtemos os geradores descritos em [2, pp. 210 e 214]. A lei de conservação correspondente a estes geradores é facilmente obtida por intermédio do nosso procedimento *Noether*:

```
> LC := Noether(L,t,x,v,%);
```

$$LC := x(t)\,t\frac{d}{dt}x(t) - t^2\left(\frac{d}{dt}x(t)\right)^2 \ln(t) = const$$

É, neste caso, muito fácil verificar a validade da lei de conservação obtida. Por definição, basta mostrar que a igualdade é verificada ao longo das extremais. A equação de Euler-Lagrange é a equação diferencial de 2^a ordem

```
> EulerLagrange(L,t,x,v);
```

$$\left\{-2\,\frac{d}{dt}x(t) - 2\,t\frac{d^2}{dt^2}x(t) = 0\right\}$$

e as extremais são as suas soluções:

```
> dsolve(%);
```

$$\{x(t) = \_C1 + \_C2\ln(t)\}$$

Substituindo as extremais na lei de conservação, obtemos, como esperado, uma proposição verdadeira:

```
> expand(subs(%,LC));
```

$$\_C2\_C1 = const$$

Exemplo 5.2 (Problema de Kepler). Analisamos agora as simetrias e leis de conservação do problema de Kepler [2, p. 217]. Neste caso o Lagrangeano tem duas variáveis dependentes ($n = 2$) e não envolve derivadas de ordem superior ($m = 1$):

$$L(t, \mathbf{q}, \dot{\mathbf{q}}) = \frac{m}{2}\left(\dot{q}_1^2 + \dot{q}_2^2\right) + \frac{K}{\sqrt{q_1^2 + q_2^2}}.$$

Vamos determinar a fórmula geral das leis de conservação. Neste caso não é possível validar a lei de conservação por aplicação directa da definição, como fizemos para o exemplo anterior, pois o Maple não é capaz de resolver o respectivo sistema de equações de Euler-Lagrange

```
> L:=m/2*(v[1]^2+v[2]^2)+K/sqrt(q[1]^2+q[2]^2);
```

[1] O sinal de percentagem (%) é um operador usado em `Maple` para referenciar o resultado do comando anterior. Para uma introdução ao `Maple`, veja-se [1].

$$L := 1/2\, m \left({v_1}^2 + {v_2}^2\right) + \frac{K}{\sqrt{{q_1}^2 + {q_2}^2}}$$

```
> EulerLagrange(L,t,[q[1],q[2]],[v[1],v[2]]);
```

$$\left\{ -m \frac{d^2}{dt^2} q_1(t) - \frac{K q_1(t)}{\left(q_1(t)^2 + q_2(t)^2\right)^{3/2}} = 0, \quad -m \frac{d^2}{dt^2} q_2(t) - \frac{K q_2(t)}{\left(q_1(t)^2 + q_2(t)^2\right)^{3/2}} = 0 \right\}$$

```
> Simetria(L, t, [q[1],q[2]], [v[1],v[2]]);
```

$$\{X_1(t, q_1, q_2) = \_C2\, q_2,\ T(t, q_1, q_2) = \_C1,\ X_2(t, q_1, q_2) = -\_C2\, q_1\}$$

```
> Noether(L, t, [q[1],q[2]], [v[1],v[2]], %):
> expand(%);
```

$$\_C2\, q_2(t)\, m \frac{d}{dt} q_1(t) - \_C2\, q_1(t)\, m \frac{d}{dt} q_2(t) - 1/2\,\_C1 \left(\frac{d}{dt} q_1(t)\right)^2 m$$
$$-1/2\,\_C1 \left(\frac{d}{dt} q_2(t)\right)^2 m + \frac{\_C1\, K}{\sqrt{q_1(t)^2 + q_2(t)^2}} = const$$

Exemplo 5.3. Vejamos o caso de um Lagrangeano com duas variáveis dependentes ($n = 2$) e com derivadas de ordem superior ($m = 2$):

$$L(t, \mathbf{x}, \dot{\mathbf{x}}, \ddot{\mathbf{x}}) = \dot{x}_1^2 + \ddot{x}_2^2$$

```
> L:=v[1]^2+a[2]^2;
```

$$L := {v_1}^2 + {a_2}^2$$

```
> Simetria(L, t, [x[1],x[2]], [v[1],v[2]], [a[1],a[2]]);
```

$$\{\ T(t, x_1, x_2) = \_C1\, t + \_C2,$$
$$X_1(t, x_1, x_2) = 1/2\,\_C1\, x_1 + \_C5,\ X_2(t, x_1, x_2) = 3/2\,\_C1\, x_2 + \_C3\, t + \_C4\ \}$$

```
> LC := Noether(L, t, [x[1],x[2]], [v[1],v[2]], [a[1],a[2]], %);
```

$$LC := \quad 2\left(1/2\,\_C1\, x_1(t) + \_C5\right) \frac{d}{dt} x_1(t) - 2\left(3/2\,\_C1\, x_2(t) + \_C3\, t + \_C4\right) \frac{d^3}{dt^3} x_2(t)$$
$$+ 2\left(\_C3 + 1/2\,\_C1 \frac{d}{dt} x_2(t)\right) \frac{d^2}{dt^2} x_2(t)$$
$$+ \left(-\left(\frac{d}{dt} x_1(t)\right)^2 - \left(\frac{d^2}{dt^2} x_2(t)\right)^2 + 2 \frac{d}{dt} x_2(t) \frac{d^3}{dt^3} x_2(t) \right) (\_C1\, t + \_C2) = const$$

Tal como para o Exemplo 5.1, também aqui é fácil verificar, por aplicação directa da definição, a validade da lei de conservação obtida:

```
> EulerLagrange(L,t, [x[1],x[2]], [v[1],v[2]], [a[1],a[2]]);
```

$$\left\{ -2 \frac{d^2}{dt^2} x_1(t) = 0, 2 \frac{d^4}{dt^4} x_2(t) = 0 \right\}$$

```
> dsolve(%);
```

$$\{\, x_1(t) = \_C1\, t + \_C2,\; x_2(t) = 1/6\, \_C3\, t^3 + 1/2\, \_C4\, t^2 + \_C5\, t + \_C6 \,\}$$

```
> expand(subs(%,LC));
```

$$2\, \_C1\, \_C5 - 3\, \_C3\, \_C1\, \_C6 + \_C1\, \_C5\, \_C4 - \_C4^2\, \_C2 + 2\, \_C3\, \_C5\, \_C2 = const$$

Exemplo 5.4 (Emden-Fowler). Consideremos o problema variacional definido pelo Lagrangeano

```
> L:= t^2/2*(v^2-(1/3)*x^6);
```

$$L := \frac{t^2\left(v^2 - \frac{x^6}{3}\right)}{2}$$

A respectiva equação diferencial de Euler-Lagrange é conhecida na astrofísica como a equação de Emden-Fowler [2, p. 220]:

```
> EL := EulerLagrange(L,t,x,v);
```

$$EL := \left\{-2\, t \frac{d}{dt} x(t) - t^2 \frac{d^2}{dt^2} x(t) - t^2 (x(t))^5 = 0\right\}$$

Encontramos os geradores infinitesimais, que conduzem a uma simetria variacional para a funcional de Emden-Fowler, por intermédio da nossa função *Simetria*:

```
> S := Simetria(L,t,x,v);
```

$$S := \left\{X(t,x) = -\frac{x\_C1}{2}, T(t,x) = \_C1\, t\right\}$$

Por exemplo,

```
> S2 := subs(_C1=-6,S);
```

$$S2 := \{T(t,x) = -6t, X(t,x) = 3x\}$$

Aplicando o Teorema de Noether (Teorema 4.1), estabelecemos a seguinte lei de conservação:

```
> simplify(Noether(L,t,x,v,S2));
```

$$t^2 \left(3\, x(t) \frac{d}{dt} x(t) + 3 \left(\frac{d}{dt} x(t)\right)^2 t + t\, (x(t))^6\right) = const$$

Exemplo 5.5 (Thomas-Fermi). Mostramos agora um exemplo de um problema do cálculo das variações que não possui nenhuma simetria variacional. Seja

```
> L:=1/2 * v^2 + 2/5 * (x^(5/2))/(sqrt(t));
```

$$L := \frac{v^2}{2} + \frac{2\, x^{\frac{5}{2}}}{5\sqrt{t}}$$

A equação de Euler-Lagrange associada a este Lagrangeano corresponde à equação diferencial de Thomas-Fermi [2, p. 220]:

```
> EL := EulerLagrange(L,t,x,v);
```

$$EL := \left\{ -\frac{d^2}{dt^2} x(t) + \frac{(x(t))^{\frac{3}{2}}}{\sqrt{t}} = 0 \right\}$$

A nossa função *Simetria* devolve, neste caso, geradores nulos. Como explicado na §3., isto significa que este problema do cálculo das variações não admite simetrias.

```
> Simetria(L, t, x, v);
```

$$\{X(t,x) = 0, T(t,x) = 0\}$$

A função *Noether* resulta num truísmo:

```
> Noether(L, t, x, v, %);
```

$$0 = const$$

Exemplo 5.6 (Oscilador Harmónico com Amortecimento). Consideremos um oscilador harmónico com força de restituição $-kx$, submerso num líquido de tal modo que o movimento da massa m é amortecido por uma força proporcional à sua velocidade. Recorrendo à segunda lei de Newton obtém-se, como equação de movimento, a equação diferencial de Euler-Lagrange associada ao seguinte Lagrangeano [10, pp. 432–434]:

```
> L:=1/2 * (m*v^2-k*x^2)*exp((a/m)*t);
```

$$L := \frac{1}{2} \left(mv^2 - kx^2\right) e^{\frac{at}{m}}$$

Para determinar um primeiro integral da equação de Euler-Lagrange, encontramos os geradores T e X sob os quais a funcional integral $J[x(\cdot)] = \int L dt$ é invariante:

```
> Simetria(L, t, x, v);
```

$$\left\{ T(t,x) = \_C1, X(t,x) = -\frac{xa\_C1}{2m} \right\}$$

```
> S:= subs(_C1=1,%);
```

$$S := \left\{ X(t,x) = -\frac{xa}{2m}, T(t,x) = 1 \right\}$$

Pelo Teorema de Noether (Corolário 4.1.2) obtemos o primeiro integral $\left(L - \dot{x}\frac{\partial L}{\partial \dot{x}}\right) T + \frac{\partial L}{\partial \dot{x}} X$:

```
> simplify(Noether(L, t, x, v, S));
```

$$-\frac{1}{2} e^{\frac{at}{m}} \left(x(t)\, a \frac{d}{dt} x(t) + m \left(\frac{d}{dt} x(t) \right)^2 + k (x(t))^2 \right) = const$$

6. Procedimentos Maple

Os procedimentos *Simetria*, *Noether* e *EulerLagrange*, descritos e ilustrados nas secções anteriores, são agora definidos usando o sistema de computação algébrica `Maple` (versão 9).

Simetria determina as simetrias de um Lagrangeano de várias variáveis dependentes e com derivadas de ordem superior, de acordo com a secção 3..

Devolve:

- conjunto/vector de geradores infinitesimais das transformações simétricas.

Forma de invocação:

- Simetria(L, t, x, x1, x2, ..., xm)

Parâmetros:

L - expressão do Lagrangeano;

t - nome da variável independente;

x - nome, lista de nomes ou vector de nomes das variáveis dependentes;

xi - nome, lista de nomes ou vector de nomes das derivadas de ordem i das variáveis dependentes;

```
Simetria:=proc(L::algebraic,t::name,x0::{name,list(name),'Vector[column]'
                       (name)},x1::{name,list(name),'Vector[column]'(name)})
  local n,m,xx,P,EqD,SysEqD,Sol,xi,Tdt,soma,V,r,i,j,k;
  if nargs<4 then print(`Nº de args insuficiente.`); return;
  elif not type([args[3..-1]],{'list'(name),'listlist'(name),
                                           'list'('Vector[column]'(name))})
    then print(`Erro na lista das var. depend. ou suas derivadas.`); return;
  end if;
  unassign('T'); unassign('X');
  xx:=convert(x0,'list')[]; n:=nops([xx]); m:=nargs-3;
  xi:=[seq(Vector(convert(args[i],'list')),i=3..m+3)];
  Tdt:=diff(T(t,xx),t)+Vector[row]([seq(diff(T(t,xx),i),i=xx)]).xi[2];
  if n>1 then P:=[Vector([seq(X[i](t,xx),i=1..n)])];
  else P:=[Vector([X(t,xx)])]; end if;
  for i from 1 to m do
    V:=Vector(n):
    for k from 0 to i-1 do
      V:=V+Matrix(n,(r,j)->diff(P[i][r], xi[k+1][j])).xi[k+2];
    end do:
    P:=[P[],map(diff,P[i],t)+V-xi[i+1]*Tdt]:
  end do:
  soma:=0:
  for i from 0 to m do
    soma:=soma+Vector[row]([seq(diff(L,r),r=convert(xi[i+1],list))]).P[i+1]:
```

```
  end do:
  EqD:=diff(L,t)*T(t,xx)+soma+L*Tdt;
  EqD:=collect(EqD,[seq(convert(xi[i+1],'list')[],i=1..m)],distributed);
  SysEqD:={coeffs(EqD,[seq(convert(xi[i+1],'list')[],i=1..m)])};
  Sol:=pdsolve(SysEqD,{T(t,xx)} union convert(P[1],'set'));
  if type(x0,'Vector') then
    return (subs(Sol,T(t,xx)),Vector(subs(Sol,P[1])));
  else return Sol; end if;
end proc:
```

Noether dados os geradores infinitesimais, determina a Lei de Conservação de um Lagrangeano de várias variáveis dependentes e com derivadas de ordem superior, de acordo com a secção 4..

Devolve:

- lei de conservação.

Formas de invocação:

- Noether(L, t, x, x1, x2, ..., xm, S)
- Noether(L, t, xs, x1s, x2s, ..., xms, T, X)

Parâmetros:

L - expressão do Lagrangeano;

t - nome da variável independente;

x - nome ou lista de nomes das variáveis dependentes;

xi - nome ou lista de nomes das derivadas de ordem i das variáveis dependentes;

S - conjunto de geradores infinitesimais das simetrias (output do procedimento *Simetria*);

xs - vector de nomes das variáveis dependentes;

xis - vector de nomes das derivadas de ordem i das variáveis dependentes;

T - gerador da transformação para a variável independente (t);

X - vector com os geradores das transformações para as variáveis dependentes (xs).

```
Noether:=proc(L::algebraic,t::name,x0::{name,list(name),'Vector[column]'
                        (name)},x1::{name,list(name),'Vector[column]'(name)})
  local xx,n,m,P,psi,LC,xi,Tdt,V,r,i,j,k;
  if type(x0,'Vector') then m:=nargs-5; else m:=nargs-4; end if;
  if m<1 then print(`N° de args insuficiente.`); return;
  elif not type([args[3..3+m]],{'list'(name),'listlist'(name),
                                        'list'('Vector[column]'(name))}) then
    print(`Erro na lista das var. depend. ou suas derivadas.`); return;
```

```
  elif (type(x0,'Vector') and not(type(args[-1],'Vector[column]') and
                  type(args[-2],algebraic))) or (not type(x0,'Vector') and
                                          not type(args[-1],'set')) then
    print(`Conj. de gerad. inválido.`); return;
  end if;
  xx:=convert(x0,'list')[];  n:=nops([xx]); unassign('T'); unassign('X');
  xi:=[seq(Vector(convert(args[i],'list')),i=3..m+3)];
  xi:=[xi[],seq(Vector([seq(x||i[k],k=1..n)]),i=m+1..2*m-1)];
  Tdt:=diff(T(t,xx),t)+Vector[row]([seq(diff(T(t,xx),i),i=xx)]).xi[2];
  if n>1 then P:=[Vector([seq(X[i](t,xx),i=1..n)])]:
  else P:=[Vector([X(t,xx)])]: end if:
  for i from 1 to (m-1) do
    V:=Vector(n):
    for k from 0 to i-1 do
      V:=V+Matrix(n,(r,j)->diff(P[i][r],xi[k+1][j])).xi[k+2] end do:
    P:=[P[],map(diff,P[i],t)+V-xi[i+1]*Tdt]:
  end do:
  psi:=[Vector[row]([seq(diff(L,i),i=convert(xi[m+1],'list'))])]:
  for i from m by -1 to 2 do
    V:=Vector[row](n):
    for k from 0 to 2*m-i do
      V:=V+LinearAlgebra[Transpose](xi[k+2]).Matrix(n,(j,r)->
                                  diff(psi[1][r],xi[k+1][j])); end do:
    psi:=[Vector[row]([seq(diff(L,i),i=convert(xi[i],'list'))])
                                         -map(diff,psi[1],t)-V,psi[]]:
  end do:
  LC:=sum('psi[i].P[i]','i'=1..m)+(L-sum('psi[i].xi[i+1]','i'=1..m))*T(t,xx)
                                                                    =const:
  if type(x0,'Vector') then
    LC:=eval(LC,[T(t,xx)=args[-2],seq(P[1][i]=args[-1][i],i=1..n)]);
  else LC:=eval(LC,args[-1]); end if;
  LC:=subs({map(i->i=i(t),[xx])[]},LC);
  LC:=subs({seq(seq(xi[k+1][i]=diff(xi[1][i](t),t$k),i=1..n),k=1..2*m-1)},
                                                                      LC);
  return LC;
end proc:
```

EulerLagrange constrói o sistema de equações de Euler-Lagrange (2.2), dado um Lagrangeano de várias variáveis dependentes e com derivadas de ordem superior.

Devolve:

- conjunto/vector de equações de Euler-Lagrange.

Forma de invocação:

- EulerLagrange(L, t, x, x1, x2, ..., xm)

Parâmetros:

- L - expressão do Lagrangeano;
- t - nome da variável independente;
- x - nome, lista de nomes ou vector de nomes das variáveis dependentes;
- xi - nome, lista de nomes ou vector de nomes das derivadas de ordem i das variáveis dependentes;

```
EulerLagrange:=proc(L::algebraic,t::name,x0::{name,list(name),
       'Vector[column]'(name)},x1::{name,list(name),'Vector[column]'(name)})
  local xx,n,m,Lxi,xi,V,EL,i,j,k;
  if nargs<4 then print(`Nº de args insuficiente.`); return;
  elif not type([args[3..-1]],{'list'(name),'listlist'(name),
                                           'list'('Vector[column]'(name))})
    then print(`Erro na lista das var. depend. ou suas derivadas.`); return;
  end if;
  xx:=convert(x0,'list')[]; n:=nops([xx]); m:=nargs-3;
  xi:=[seq(Vector(convert(args[i],'list')),i=3..m+3)];
  V:=[0$n];
  for i from 1 to m do
    Lxi:=[seq(diff(L,k),k=convert(xi[i+1],'list'))]:
    Lxi:=subs({map(k->k=k(t),[xx])[]},Lxi);
    Lxi:=subs({seq(seq(xi[k+1][j]=diff(xi[1][j](t),t$k),j=1..n),k=1..m)},
                                                                      Lxi);
    V:=V+(-1)^i*map(diff,Lxi,t$i);
  end do:
  EL:=[seq(diff(L,k),k=convert(xi[1],'list'))];
  EL:=subs({map(k->k=k(t),[xx])[]},EL);
  EL:=subs({seq(seq(xi[k+1][j]=diff(xi[1][j](t),t$k),j=1..n),k=1..m)},EL);
  EL:=EL+V;
  if type(x0,'Vector') then return convert(map(i->i=0,EL),'Vector[column]');
  elif type(x0,'list') then return convert(map(i->i=0,EL),'set');
  else return op(EL)=0; end if;
end proc:
```

7. Conclusão e Trabalho Futuro

Os sistemas actuais de computação algébrica colocam à nossa disposição ambientes de computação científica extremamente sofisticados e poderosos. Disponibilizam já muito conhecimento matemático e permitem estender esse conhecimento por intermédio de linguagens de programação de muito alto nível, expressivas e intuitivas, próximas da linguagem matemática. Tais ambientes permitem a realização de uma miríade de cálculos matemáticos simbólicos, com extrema eficiência, e a definição célere de novas funcionalidades. Neste trabalho usámos o sistema de computação matemática `Maple 9` para definir novas funções que permitem a determinação automática de simetrias e leis de conservação no cálculo das variações. Mostrámos depois, por meio de exemplos concretos, como as novas funcionalidades são de grande utilidade prática. Como trabalho futuro pretendemos estender o nosso "package Maple" ao caso discreto e aos problemas mais genéricos tratados pelo controlo óptimo.

Uma abordagem análoga à que fizemos pode ser efectuada para problemas do cálculo das variações discretos no tempo. No caso discreto, o problema fundamental do cálculo das variações consiste em determinar uma sequência finita $\mathbf{x}(k) \in \mathbb{R}^n$, $k = M, \ldots, M + N$, de modo a que a função de custo discreta

$$J[\mathbf{x}(\cdot)] = \sum_{k=M}^{M+N-1} L(k, \mathbf{x}(k), \mathbf{x}(k+1))$$

seja minimizada (ou maximizada). Neste tipo de problemas consideramos um intervalo de tempo de N períodos, com início num período fixo M, em que $k \in \mathbb{Z}$, $k = M, \ldots, M+N-1$, é a variável discreta tempo, e assumimos que o Lagrangeano L é continuamente diferenciável em $\{M, \ldots, M+N-1\} \times \mathbb{R}^n \times \mathbb{R}^n$. À semelhança do que fizemos para o caso contínuo, é natural considerarem-se problemas de optimização em que o Lagrangeano envolve diferenças finitas de ordem superior,

$$J[\mathbf{x}(\cdot)] = \sum_{k=M}^{M+N-1} L(k, \mathbf{x}(k), \mathbf{x}(k+1), \ldots, \mathbf{x}(k+m)), \tag{7.1}$$

com $m \geq 1$, e considerando ainda o Lagrangeno continuamente diferenciável relativamente a todos os seus argumentos. Uma condição necessária para que $\mathbf{x}(k)$ seja uma extremal de (7.1) é dada pela equação de Euler-Lagrange discreta

$$\sum_{j=0}^{m} \frac{\partial L}{\partial \mathbf{x}^j}(k+m-j, \mathbf{x}(k+m-j), \ldots, \mathbf{x}(k+2m-j)) = \mathbf{0}\,. \tag{7.2}$$

No caso $m = 1$, a equação (7.2) reduz-se à equação discreta de Euler-Lagrange

$$\frac{\partial L}{\partial \mathbf{x}}(k+1, \mathbf{x}(k+1), \mathbf{x}(k+2)) + \frac{\partial L}{\partial \mathbf{x}^1}(k, \mathbf{x}(k), \mathbf{x}(k+1)) = \mathbf{0}.$$

Definição 7.1. *Um problema discreto expresso pela função de custo (7.1) diz-se invariante sob as transformações uni-paramétricas $\mathbf{h}^s(k, \mathbf{x})$, com $\mathbf{h}^0(k, \mathbf{x}) = \mathbf{x}$, se, para todo o s suficientemente pequeno e qualquer que seja o k,*

$$L(k, \mathbf{x}(k), \mathbf{x}(k{+}1), \ldots, \mathbf{x}(k{+}m)) = L(k, \mathbf{h}^s(k, \mathbf{x}(k)), \ldots, \mathbf{h}^s(k{+}m, \mathbf{x}(k{+}m))). \tag{7.3}$$

A partir desta definição conseguimos estabelecer uma condição necessária e suficiente de invariância.

Teorema 7.1. *A função de custo (7.1) é invariante sob as transformações uni-paramétricas $\mathbf{h}^s(k, \mathbf{x})$, com geradores infinitesimais $\mathbf{X}(k, \mathbf{x}) = \frac{\partial}{\partial s}\mathbf{h}^s(k, \mathbf{x})\big|_{s=0}$, se, e apenas se,*

$$\sum_{j=0}^{m} \frac{\partial L}{\partial \mathbf{x}^j} \cdot \mathbf{X}(k+j, \mathbf{x}(k+j)) = 0, \tag{7.4}$$

onde o Lagrangeano discreto L é avaliado em $(k, \mathbf{x}(k), \mathbf{x}(k+1), \ldots, \mathbf{x}(k+m))$.

Observação 7.1. *Quando $m = 1$, a equação (7.4) reduz-se a*

$$\frac{\partial L}{\partial \mathbf{x}} \cdot \mathbf{X}(k, \mathbf{x}(k)) + \frac{\partial L}{\partial \mathbf{x}^1} \cdot \mathbf{X}(k+1, \mathbf{x}(k+1)) = 0.$$

Demonstração (Teorema 7.1). Da condição de invariância (7.3) podemos escrever:

$$\frac{\mathrm{d}}{\mathrm{d}s} L(k, \mathbf{h}^s(k, \mathbf{x}(k)), \ldots, \mathbf{h}^s(k+m, \mathbf{x}(k+m))) = 0\,.$$

Derivando, obtemos, para $s = 0$, a igualdade (7.4):

$$\sum_{j=0}^{m} \frac{\partial L}{\partial \mathbf{x}^j} \cdot \frac{\partial}{\partial s} \mathbf{h}^s(k+j, \mathbf{x}(k+j)) \bigg|_{s=0} = 0$$

com L avaliado em $(k, \mathbf{x}(k), \mathbf{x}(k+1), \ldots, \mathbf{x}(k+m))$.

□

É possível deduzir um teorema análogo ao teorema de Noether, para o caso discreto, que permite estabelecer leis de conservação discretas (cf. [14]).

Teorema 7.2 (Teorema de Noether Discreto). *Se a função de custo (7.1) é invariante, no sentido da definição 7.1, sob as transformações uni-paramétricas $\mathbf{h}^s(k, \mathbf{x})$, com geradores infinitesimais $\mathbf{X}(k, \mathbf{x}) = \frac{\partial}{\partial s} \mathbf{h}^s(k, \mathbf{x})\big|_{s=0}$, então todas as soluções $\mathbf{x}(k)$ da equação de Euler-Lagrange (7.2) satisfazem*

$$\sum_{j=0}^{m-1} \mathbf{\Psi}^j(k) \cdot \mathbf{X}(k+j, \mathbf{x}(k+j)) = const \tag{7.5}$$

onde

$$\mathbf{\Psi}^0(k) = \frac{\partial L}{\partial \mathbf{x}}(k, \mathbf{x}(k), \ldots, \mathbf{x}(k+m))$$
$$\mathbf{\Psi}^j(k) = \mathbf{\Psi}^{j-1}(k+1) + \frac{\partial L}{\partial \mathbf{x}^j}(k, \mathbf{x}(k), \ldots, \mathbf{x}(k+m)), \quad para\ j = 1, 2, \ldots, m-1\,.$$

Observação 7.2. *Caso $m = 1$, a lei de conservação (7.5) reduz-se a*

$$\frac{\partial L}{\partial \mathbf{x}}(k, \mathbf{x}(k), \mathbf{x}(k+1)) \cdot \mathbf{X}(k, \mathbf{x}(k)) = const.$$

Para problemas discretos no tempo, facilmente se definem em Maple procedimentos que implementem as condições (7.2), (7.4) e (7.5).

O controlo óptimo pode ser encarado como uma extensão natural do cálculo das variações [7, 8]. A resolução de problemas do controlo óptimo passa normalmente pela aplicação do Princípio do Máximo de Pontryagin, que constitui uma generalização das condições clássicas de Euler-Lagrange e de Weierstrass do cálculo das variações. Em termos práticos, algorítmicos, temos de resolver um sistema de equações diferenciais, chamado de sistema Hamiltoniano, depois de eliminados os parâmetros de controlo por intermédio da *condição de máximo*. Tal como no cálculo das variações, também no controlo óptimo as equações diferenciais ordinárias obtidas são em geral não-lineares e de difícil resolução (podem, inclusive, não ser integráveis). As leis de conservação são usadas para simplificar essas equações e, mais uma vez, a questão que se coloca é saber como determinar tais quantidades preservadas. Resulta que os resultados clássicos de Emmy Noether podem ser generalizados ao contexto do controlo óptimo, reduzindo o problema ao da descoberta de grupos de transformações uni-paramétricas que deixem o problema de controlo óptimo invariante [13]. Seria de grande utilidade prática dispor de facilidades computacionais que permitissem identificar as simetrias dos problemas de controlo óptimo [5, 6].

Serão estas algumas das direcções do nosso trabalho futuro.

Referências

[1] L. N. de Andrade, "Introdução à Computação Algébrica com o Maple", IMPA, editora da Sociedade Brasileira de Matemática, 2004.

[2] B. van Brunt, "The Calculus of Variations", Springer-Verlag New York, 2004.

[3] G. Buttazzo and M. Belloni, A Survey on Old and Recent Results About the Gap Phenomenon in the Calculus of Variations, *Recent developments in well-posed variational problems*, Kluwer Acad. Publ., 1995, 1–27.

[4] F. S. David, "O Cálculo Variacional Clássico e Algumas das suas Aplicações à Física Matemática – Referência a Algumas Extensões mais Recentes", Electricidade de Portugal, 1986.

[5] J. W. Grizzle and S. I. Marcus, The structure of nonlinear control systems possessing symmetries, *IEEE Trans. Automat. Control*, Vol. 30, No. 3 (1985), 248–258.

[6] A. Gugushvili, O. Khutsishvili, V. Sesadze, G. Dalakishvili, N. Mchedlishvili, T. Khutsishvili, V. Kekenadze and D. F. M. Torres, "Symmetries and Conservation Laws in Optimal Control Systems", Georgian Technical University, Tbilisi, 2003.

[7] L. P. Lebedev and M. J. Cloud, "The Calculus of Variations and Functional Analysis – with Optimal Control and Applications in Mechanics", World Scientific, 2003.

[8] A. Leitão, "Cálculo Variacional e Controle Ótimo", IMPA, 23^o Colóquio Brasileiro de Matemática, 2001.

[9] J. D. Logan, "Invariant Variational Principles", Academic Press [Harcourt Brace Jovanovich Publishers], 1977.

[10] J. D. Logan, "Applied Mathematics – A Contemporary Approach", John Wiley & Sons, New York, 1987.

[11] E. Noether, Invariante Variationsprobleme, *Gött. Nachr.* (1918), 235–257.

[12] P. J. Olver, "Applications of Lie Groups to Differential Equations", Springer-Verlag, 1986.

[13] D. F. M. Torres, On the Noether Theorem for Optimal Control, *European Journal of Control*, Vol. 8, No. 1 (2002), 56–63.

[14] D. F. M. Torres, Integrals of Motion for Discrete-Time Optimal Control Problems, *Control Applications of Optimisations 2003*, R. Bars and E. Gyurkovics eds., IFAC Workshop Series (2003), 33–38.

[15] D. F. M. Torres, Proper Extensions of Noether's Symmetry Theorem for Nonsmooth Extremals of the Calculus of Variations, *Communications on Pure and Applied Analysis*, Vol. 3, No. 3 (2004), 491–500.